International Amateur Radio Union Region 1

Europe, Middle East, Africa and Northern Asia Founded 1950

General Conference, Davos, 11 to 16 September 2005

SUBJECT	**DX Records to be based on WGS 84**		
Society	**EDR**	**Country:**	**Denmark**
Committee:	**C5**	**Paper number:**	**07**
Contact:	**Ivan G. Stauning, OZ7IS**	**e-mail:**	**is@ihk.dk**

Introduction

This document addresses the issue of the discrepancy between the way a radio station's coordinates are found (WGS 84) and the way a DX record distance is calculated (Datum Hayes 1924).

Background

The IARU Region 1 VHF Committee decided to recommend (Lillehammer 1999) that station coordinates are based upon the WGS 84 Earth Model.
However, IARU Region 1 VHF Committee still uses the "GM4ANB formula" that uses the Datum Hayes 1924 Earth Model.
As a consequence there is a discrepancy between the reference model that is used when finding a station's location and the reference model used when calculating distances between locations.

Key points and proposal

1. Since IARU Region 1 VHF Committee recommends determining a station location based upon WGS 84 the DX Records should also be calculated using the WGS 84 Earth Model; replacing the Datum Hayes 1924 Earth Model.

2. The formula used to calculate the distance between two stations, for DX Record QSOs, is the Vincenty's Inverse Algorithm, as shown below, in this case Delphi Pascal.

```
Procedure DistDirs_WGS84(La1, Lo1, La2, Lo2:double; var Dist, Dir12, Dir21:double);
Function ModCrs(x:extended):extended;
begin
  ModCrs:=x-2*Pi*Floor(x/(2*Pi));
end;
const
  a=6378.137;              // a and 1/f WGS 84 original defaults
  f=1/298.257223563;
  MaxI=100;                // Max number of integral iterations
var
  I:integer;
  r,tu1,tu2,cu1,su1,cu2,s1,b1,f1,x,
  sx,cx,sy,cy,y,sa,c2a,cz,e,c,d:double;
begin
  If ((La1=La2) and (Lo1=Lo2)) then
  begin // Pos1 and Pos2 are identical
    Dist:=0;
    Dir12:=0;
    Dir21:=0;
    Exit;
  end;
  If ((Abs(La1)=90) and (Abs(La2)=90)) then
```

```
   La1:=La1+0.00001;       // allow algorithm to complete
  La1:=La1*DtR;
  Lo1:=Lo1*DtR;
  La2:=La2*DtR;
  Lo2:=Lo2*DtR;
  I:=1;
  r:=1-f;
  tu1:=r*Tan(La1);
  tu2:=r*Tan(La2);
  cu1:=1/Sqrt(1+tu1*tu1);
  su1:=cu1*tu1;
  cu2:=1/Sqrt(1+tu2*tu2);
  s1:=cu1*cu2;
  b1:=s1*tu2;
  f1:=b1*tu1;
  x:=Lo2-Lo1;
  d:=x+1;                  // force one pass
  While ((Abs(d-x)>NumPre) AND (I<MaxI)) do
  begin
    Inc(I);
    sx:=Sin(x);
    cx:=Cos(x);
    tu1:=cu2*sx;
    tu2:=b1-su1*cu2*cx;
    sy:=Sqrt(tu1*tu1+tu2*tu2);
    cy:=s1*cx+f1;
    y:=ArcTan2(sy,cy);
    sa:=s1*sx/sy;
    c2a:=1-sa*sa;
    cz:=f1+f1;
    If c2a>0 then
      cz:=cy-cz/c2a;
    e:=cz*cz*2-1;
    c:=((-3*c2a+4)*f+4)*c2a*f/16;
    d:=x;
    x:=((e*cy*c+cz)*sy*c+y)*sa;
    x:=(1-c)*x*f+Lo2-Lo1;
  end;
  Dir12:=ModCrs(ArcTan2(tu1,tu2))*180/Pi;
  Dir21:=ModCrs(ArcTan2(cu1*sx,b1*cx-su1*cu2)+Pi)*180/Pi;
  x:=Sqrt((1/(r*r)-1)*c2a + 1);
  X:=X+1;
  x:=(x-2)/x;
  c:=1-x;
  c:=(x*x/4+1)/c;
  d:=(0.375*x*x-1)*x;
  x:=e*cy;
  Dist:=((((sy*sy*4-3)*(1-e-e)*cz*d/6-x)*d/4+cz)*sy*d+y)*c*a*r;
  If Abs(I-MaxI)<NumPre then
  begin
    Dist:=-1;
    Dir12:=-1;
    Dir21:=-1;
  end;
end;
```

The DX Record Distance is mathematically rounded to an integer value.

Recommendations

1. *There must be coherence between the way a station's location is found and the earth model used when calculating DX record distances*

2. Vincenty's Inverse Algorithm is used for calculating DX record distances.

3. The WGS 84 earth model with a = 6378,137 and f = 1/298,257223563 is used.

4. All existing DX record distance QSOs are recalculated accordingly.